/*

Nayak 609
Proposal title: Randomized Drug Trials without Consent? The Public’s Perspective

HYPOTHESES

Stated-Hyp 1: We hypothesize that the first scenario will be perceived as riskier because the drug subjects receive (as opposed to the timing of drug administration) will be chosen randomly.

	Test-Hyp 1: Participants in the first scenario & GN condition will differ in their support from participants in the second scenario & GN condition.
	
	Test-Hyp 2: Participants in the first scenario & VC condition will differ in their support from participants in the second scenario & VC condition.

Stated-Hyp 2: We propose this between-subject methodology to test whether there is preference for the GN model or VC model in order to minimize anchoring or framing effects that could arise when directly comparing the GN and VC models

	Test-Hyp 3: Participants in the GN & first scenario will differ in their support from participants in the VC & first scenario condition.
	
	Test-Hyp 4: Participants in the GN & second scenario will differ in their support from participants in the VC & second scenario condition.

Stated-Hyp 3: Does the level of risk of the research study influence support for either the GN or VC models?

	Test-Hyp 5: Perceived level of risk will be associated with support among participants in the GN condition.
	
	Test-Hyp 6: Perceived level of risk will be associated with support among participants in the VC condition.

********************************************************************************

NOTES: 

1. Control variables
We did not control for participants's prior history with high blood pressure because we could not locate these variables in the dataset.
Did not control for demographics because it was unclear specifically which demographics should be controlled for.

2. Recoding strategy
Dichotimized the primary DV because the authors wrote "We plan to dichotomize the Likert-scaled outcome variable."

3. Analysis strategy
Used regression analyses to be consistent with our general approach. The authors wrote that they would use "independent samples (t-test)". Regression analyses provide t-tests.

4. Additional hypotheses
There are additional potential hypotheses. We decided not to test these potential hypotheses because they were worded more as exploratory analyses and - if included - would outweigh the more clearly specified hypotheses. Specifically, the authors wrote
- "We are interested in seeing if these views (value of the study; risk of participating; benefit of participating; importance of disclosing how the treatment is chosen in randomized trials; importance of disclosing how the treatment is chosen in usual care) are predictive (within a study arm) of the respondent’s opinion of conducting the study without written informed consent, in addition to demographic characteristics."
- "We are also interested in testing whether there are differences in the
responses to these questions between study arms."
- "It is important to test whether the respondent’s history and familiarity with high blood pressure affects their responses to the survey."
*/

clear all
use Nayak609.dta, clear

********************************************************************************

* RECODING

* Experimental Conditions
	tab xtess175, mis
	tab xtess175, nolabel
	tab dov_option, mis
	tab dov_option, nolabel
	gen scenarioCondition = 0
	gen consentCondition = 0
	replace scenarioCondition = 1 if xtess175 == 2
	replace consentCondition = 1 if dov_option == 2
	label define scenarioCondition_l 0 "Scenario 1" 1 "Scenario 2"
	label define consentCondition_l 0 "GN" 1 "VC"
	label values scenarioCondition scenarioCondition_l  
	label values consentCondition consentCondition_l  
	tab scenarioCondition xtess175, mis
	tab consentCondition dov_option, mis

* Risk Perception
	tab scenarioCondition q4a, nolab mis
	tab scenarioCondition q4b, nolab mis
	gen riskPerception = .
	replace riskPerception = q4a if scenarioCondition == 0
	replace riskPerception = q4b if scenarioCondition == 1
	replace riskPerception = . if riskPerception == -1
	tab riskPerception, mis
	
* DV: Support
	tab q1a, mis
	tab q1b, mis
	tab q1c, mis
	tab q1d, mis
	gen advice = .
	replace advice = q1a if xtess175 == 1 & dov_option == 1
	replace advice = q1b if xtess175 == 1 & dov_option == 2
	replace advice = q1c if xtess175 == 2 & dov_option == 1
	replace advice = q1d if xtess175 == 2 & dov_option == 2
	replace advice = . if advice == -1
	tab advice, mis nolabel
	
	tab q2a, mis
	tab q2b, mis
	tab q2c, mis
	tab q2d, mis
	gen preference = .
	replace preference = q2a if xtess175 == 1 & dov_option == 1
	replace preference = q2b if xtess175 == 1 & dov_option == 2
	replace preference = q2c if xtess175 == 2 & dov_option == 1
	replace preference = q2d if xtess175 == 2 & dov_option == 2
	replace preference = . if preference == -1
	tab preference, mis nolabel

	gen support = (advice + preference) / 2
	gen support_dic = 0
	replace support_dic = 1 if support > 2.5
	replace support_dic = . if support == .
	tab support support_dic, mis

********************************************************************************

* ANALYSIS

* Test-Hyp 1: Participants in the first scenario & GN condition will differ in their support from participants in the second scenario & GN condition.
	ttest support_dic if consentCondition == 0, by(scenarioCondition) 
	// no support for H1 (p = .053)
	tess, ttest init(Nayak609)
	
* Test-Hyp 2: Participants in the first scenario & VC condition will differ in their support from participants in the second scenario & VC condition.
	ttest support_dic if consentCondition == 1, by(scenarioCondition) 
	// no support for H2 (p = .902)
	tess, ttest
	
* Test-Hyp 3: Participants in the GN & first scenario will differ in their support from participants in the VC & first scenario condition.
	ttest support_dic if scenarioCondition == 0, by(consentCondition) 
	// support for H3 (p = .011)
	tess, ttest
	
* Test-Hyp 4: Participants in the GN & second scenario will differ in their support from participants in the VC & second scenario condition.
	ttest support_dic if scenarioCondition == 1, by(consentCondition) 
	// no support for H4 (p = .635)
	tess, ttest

* Test-Hyp 5: Perceived level of risk will be association for support among participants in the GN condition.
	reg support_dic riskPerception if consentCondition == 0
	// support for H5 (p = .000)
	tess riskPerception
	
* Test-Hyp 6: Perceived level of risk will be association for support among participants in the VC condition.
	reg support_dic riskPerception if consentCondition == 1
	// support for H6 (p = .000)
	tess riskPerception	
